SUPPLY CHAIN ANALYTICS

Excel Project Report

Prepared by	Larry Vu
Dataset	Supply Chain Dataset - 100 SKUs, 24 variables
Tools	Microsoft Excel
Date	May 2026

Table of Contents

1. Executive Summary

This report documents the design, methodology, and findings of a supply chain analytics project built entirely in Microsoft Excel. The project transforms a 100-SKU, 24-variable synthetic supply chain dataset into a structured, formula-driven analytical workbook comprising eight sheets - five analysis pivot tables, two reporting templates, and a formatted raw data layer.

The workbook was developed to demonstrate advanced Excel competency aligned with professional data analyst requirements: multi-criteria aggregation formulas, derived KPI calculations, conditional formatting, embedded charts, and cross-sheet live references. All analytical outputs update automatically as source data changes.

Avg Defect Rate	Inspection Pass Rate	Avg Availability
2.28%	23%	48.4%

Total Revenue	Total Units Sold	SKUs Analysed
$577,605	46,099	100

Three critical performance issues were identified: a 23% inspection pass rate (well below acceptable benchmarks), average stock availability of 48.4% against a 60% target, and Supplier 4 recording zero passing inspections across all 18 of its SKUs.

2. Project Objectives

The project was designed to address the following analytical and technical objectives:

2.1. Analytical Objectives

- Identify revenue and sales trends across product categories, customer segments, and geographic locations
- Evaluate supplier performance across cost, lead time, production volume, and quality dimensions
- Assess logistics efficiency by carrier, transportation mode, and route
- Profile inventory health through stock levels, availability rates, and order quantities by location
- Quantify quality performance via defect rates and inspection outcomes across products and suppliers

2.2. Technical Objectives

- Build a formula-driven workbook that auto-recalculates from a single Raw Data source table
- Demonstrate advanced Excel functions: SUMIF, AVERAGEIF, COUNTIF, COUNTIFS, and IFERROR
- Create derived KPI metrics not available as native pivot aggregations (pass rates, revenue share, stock-to-order ratios)
- Apply conditional formatting (colour scales and data bars) to surface performance patterns visually
- Embed charts directly within analysis sheets to support at-a-glance interpretation
- Produce professional reporting templates suitable for recurring business use

3. Dataset Overview

The source dataset is a synthetic supply chain dataset containing 100 SKU records across 24 variables, spanning product details, supplier information, logistics data, inventory metrics, sales figures, and quality outcomes.

Kaggle link: https://www.kaggle.com/datasets/litonislam/supply-chain-data-analysis

3.1. Dataset Dimensions

Dimension	Attribute	Distinct Values
Product	Product type	Cosmetics (26), Haircare (34), Skincare (40)
Supplier	Supplier name	Suppliers 1-5 (15-27 SKUs each)
Location	Warehouse city	Bangalore, Chennai, Delhi, Kolkata, Mumbai
Logistics	Shipping carriers	Carrier A (28), Carrier B (43), Carrier C (29)
Logistics	Transportation modes	Air (26), Rail (28), Road (29), Sea (17)
Logistics	Routes	Route A (43), Route B (37), Route C (20)
Quality	Inspection results	Pass (23), Fail (36), Pending (41)
Customer	Demographics	Female, Male, Non-binary, Unknown

3.2. Key Variables

Variable	Type	Description
SKU	ID	Unique product identifier
Product type	Categorical	Product category
Price	Numeric ($)	Unit selling price
Availability	Numeric (%)	Product availability percentage
Number of products sold	Integer	Sales volume
Revenue generated	Numeric ($)	Total revenue per SKU
Stock levels	Integer	Units held in warehouse
Lead times	Integer (days)	Operational fulfilment lead time
Lead time	Integer (days)	Supplier-quoted lead time
Order quantities	Integer	Units ordered per cycle
Shipping times	Integer (days)	Transit time to customer
Shipping costs	Numeric ($)	Carrier fee per shipment
Costs	Numeric ($)	Total logistics cost (incl. handling)
Manufacturing costs	Numeric ($)	Per-unit production cost
Manufacturing lead time	Integer (days)	Time to manufacture one batch
Defect rates	Numeric (%)	Percentage of defective units
Inspection results	Categorical	Quality outcome: Pass / Fail / Pending
Production volumes	Integer	Total units produced per supplier SKU

4. Workbook Architecture

The workbook follows a single-source-of-truth architecture: all 337 formulas across the five analysis sheets reference one named Excel Table (RawData) in the Raw Data sheet. Changing any value in the Raw Data sheet propagates through every analysis table, chart, and KPI card automatically.

Sheet	Type	Purpose	Formulas
Raw Data	Source	Formatted Excel Table - all 100 rows × 24 columns	4
Revenue by Product	Analysis	Revenue, units, pricing, and defect by product category	~48
Supplier Performance	Analysis	Lead time, cost, defect rate, pass/fail by supplier	~63
Logistics & Shipping	Analysis	Three sub-tables: carrier, mode, and route	~72
Location & Inventory	Analysis	Stock, availability, revenue, and derived ratios by city	~72
Quality & Defects	Analysis	Defect rates and inspection counts by product and supplier	~78
KPI Report Template	Template	14-row live KPI dashboard with traffic-light status	~14
Monthly Report Template	Template	Blank structured monthly report with variance formulas	~36

4.1. Raw Data Table

The source data is stored as a named Excel Table (RawData), which enables structured column references in formulas (e.g. RawData[Product type]) rather than range references. This means formulas remain correct even if rows are added or reordered, and the table formatting - alternating row colours, frozen panes at column C, row 2 - makes navigation intuitive at scale.

5. Analysis Sheets - Pivot Methods & Formulas

Each analysis sheet uses Excel's lookup and aggregation functions to replicate pivot table behaviour while adding derived columns that native pivot tables cannot produce. The core pattern is: group by a categorical dimension, aggregate numeric fields using SUMIF/AVERAGEIF/COUNTIF, then derive calculated fields from those aggregations.

5.1. Revenue by Product

This sheet groups all 100 SKUs by product type (Cosmetics, Haircare, Skincare) to compare revenue contribution, sales volume, pricing, and quality across categories.

Pivot Structure

Field	Role	Aggregation
Product type	Row dimension	GROUP BY
SKU	Count column	COUNTIF
Number of products sold	Value column	SUMIF
Revenue generated	Value column	SUMIF
Revenue share %	Derived column	Group revenue / SUM(all revenue)
Price	Value column	AVERAGEIF

Defect rates	Value column	AVERAGEIF / 100
Lead times	Value column	AVERAGEIF

Formula Reference

Cell	Formula
B5	`=COUNTIF(RawData[Product type], A5)`
C5	`=SUMIF(RawData[Product type], A5, RawData[Number of products sold])`
D5	`=SUMIF(RawData[Product type], A5, RawData[Revenue generated])`
E5	`=D5 / SUM($D$5:$D$7)   ← denominator locked with $ to prevent drift when copied`
F5	`=AVERAGEIF(RawData[Product type], A5, RawData[Price])`
G5	`=AVERAGEIF(RawData[Product type], A5, RawData[Defect rates]) / 100`
H5	`=AVERAGEIF(RawData[Product type], A5, RawData[Lead times])`
Totals B	`=SUM(B5:B7)`
Totals D	`=SUM(D5:D7)   ← grand total revenue`
Totals E	`=SUM(E5:E7)   ← must equal 100%`

Key Findings

- Skincare is the top revenue driver: 40 SKUs generating $241,628 (41.8% of total revenue)
- Haircare holds the highest average defect rate at 2.48% despite being the second-largest category
- Cosmetics achieves the lowest defect rate (1.92%) with the smallest SKU count (26)
- Revenue share uses a locked denominator (SUM($D$5:$D$7)) so the formula copies correctly across rows without shifting the range

5.2. Supplier Performance

This sheet ranks all five suppliers across eight performance dimensions. The key design decision is to use AVERAGEIF (not SUMIF) for cost and lead time metrics, since these are rates that would be inflated by suppliers with larger SKU counts if summed.

Pivot Structure

Field	Role	Aggregation	Why avg not sum?
Supplier name	Row dimension	GROUP BY	-
SKU	Count	COUNTIF	-
Lead time	Quality metric	AVERAGEIF	Rate - summing inflates by SKU count
Manufacturing lead time	Quality metric	AVERAGEIF	Rate
Manufacturing costs	Cost metric	AVERAGEIF	Per-unit rate
Defect rates	Quality KPI	AVERAGEIF / 100	Rate
Production volumes	Capacity metric	SUMIF	Cumulative output
Inspection Pass Rate	Derived KPI	COUNTIFS / COUNTIF	Two-criteria - not native pivot
Inspection Fail Rate	Derived KPI	COUNTIFS / COUNTIF	Two-criteria - not native pivot

B5	=COUNTIF(RawData[Supplier name], A5)
C5	=AVERAGEIF(RawData[Supplier name], A5, RawData[Lead time])
D5	=AVERAGEIF(RawData[Supplier name], A5, RawData[Manufacturing lead time])
E5	=AVERAGEIF(RawData[Supplier name], A5, RawData[Manufacturing costs])
F5	=AVERAGEIF(RawData[Supplier name], A5, RawData[Defect rates]) / 100
G5	=SUMIF(RawData[Supplier name], A5, RawData[Production volumes])
H5	=COUNTIFS(RawData[Supplier name], A5, RawData[Inspection results], "Pass") / COUNTIF(RawData[Supplier name], A5)
I5	=COUNTIFS(RawData[Supplier name], A5, RawData[Inspection results], "Fail") / COUNTIF(RawData[Supplier name], A5)

The COUNTIFS function is the critical formula here. A native pivot table can count all inspection results, but cannot split counts by outcome without a calculated field. COUNTIFS takes two criteria - the supplier name in column A and the specific inspection text ("Pass" or "Fail") - allowing precise conditional counts per supplier.

Key Findings

- Supplier 1 is the best performer: lowest defect rate (1.80%), shortest lead time (14.8 days), and 13 of 27 SKUs passing inspection (48.1%)
- Supplier 4 is a critical risk: zero passing inspections across all 18 SKUs, highest manufacturing cost ($62.71/unit), second-lowest defect rate (2.34%) - suggesting the defect rate metric alone underrepresents the quality issue
- Supplier 5 has the highest defect rate (2.67%) and the second-lowest pass rate
- Supplier 3 has the longest lead time (20.1 days), exceeding the 15-day target by 34%

5.3. Logistics & Shipping

This sheet contains three independent pivot sub-tables in a single worksheet - by carrier, by transport mode, and by route. Each uses a different grouping field but the same core formula pattern.

Important Column Distinction

Two cost columns exist in the dataset and are used differently across the sub-tables:

- Shipping costs: the carrier fee only - used in the By Carrier table
- Costs: total logistics cost including handling, warehousing, and carrier fees - used in the By Mode and By Route tables for a complete cost picture

Formula Reference - By Carrier

Cell	Formula
B6	=COUNTIF(RawData[Shipping carriers], A6)
C6	=AVERAGEIF(RawData[Shipping carriers], A6, RawData[Shipping times])
D6	=AVERAGEIF(RawData[Shipping carriers], A6, RawData[Shipping costs])
E6	=SUMIF(RawData[Shipping carriers], A6, RawData[Shipping costs])

Formula Reference - By Transport Mode

Cell	Formula
H6	=COUNTIF(RawData[Transportation modes], G6)

I6	`=AVERAGEIF(RawData[Transportation modes], G6, RawData[Costs])  ← uses Costs, not Shipping costs`
J6	`=SUMIF(RawData[Transportation modes], G6, RawData[Costs])`
K6	`=AVERAGEIF(RawData[Transportation modes], G6, RawData[Defect rates]) / 100`

Formula Reference - By Route

Cell	Formula
B14	`=COUNTIF(RawData[Routes], A14)`
C14	`=AVERAGEIF(RawData[Routes], A14, RawData[Costs])`
D14	`=SUMIF(RawData[Routes], A14, RawData[Revenue generated])`
E14	`=SUMIF(RawData[Routes], A14, RawData[Costs]) / D14  ← derived: Cost-to-Revenue ratio`

Key Findings

- Carrier B handles the most shipments (43) at the lowest average shipping time (5.3 days) - the most efficient carrier
- Air transport is the most expensive mode ($561.71 avg total cost) but carries the lowest defect rate (1.82%) - suggesting it is used for higher-value, more sensitive products
- Sea transport is the cheapest mode ($417.82 avg total cost) but handles the fewest shipments (17)
- Route A carries the most volume (43 shipments, $253,199 revenue) at the lowest average cost - the preferred route
- Route B has the highest average cost ($595.66) despite carrying fewer shipments - a candidate for route optimisation

5.4. Location & Inventory

This sheet aggregates inventory and operational metrics by warehouse city. Two derived columns extend the standard aggregations to provide actionable inventory management signals.

Lead Time Column Choice

The dataset contains two lead time columns: Lead time (supplier-quoted) and Lead times (operational fulfilment). This sheet uses Lead times because it reflects actual warehouse-to-customer performance, which is the relevant metric for inventory planning.

Formula Reference

Cell	Formula
B5	`=COUNTIF(RawData[Location], A5)`
C5	`=AVERAGEIF(RawData[Location], A5, RawData[Stock levels])`
D5	`=AVERAGEIF(RawData[Location], A5, RawData[Availability]) / 100`
E5	`=AVERAGEIF(RawData[Location], A5, RawData[Lead times])  ← operational, not supplier`
F5	`=SUMIF(RawData[Location], A5, RawData[Revenue generated])`
G5	`=AVERAGEIF(RawData[Location], A5, RawData[Order quantities])`
H5	`=IFERROR(C5 / G5, "-")  ← derived: Stock-to-Order ratio`
I5	`=IFERROR(F5 / B5, "-")  ← derived: Revenue per SKU`

IFERROR wraps both derived formulas to prevent #DIV/0! errors if a location has no SKUs in a filtered view. The "-" fallback keeps the table readable.

Derived Metrics Explained

Metric	Formula	Business Meaning
Stock-to-Order Ratio	Avg Stock / Avg Order Qty	Ratio < 1 means stock on hand is less than a typical order - high stockout risk
Revenue per SKU	Total Revenue / SKU count	Measures how productively each SKU generates revenue in that location

Key Findings

- Mumbai generates the most revenue ($137,755) with the fewest SKUs per dollar - highest revenue-per-SKU efficiency
- Kolkata has the best average stock level (57.6 units) but the lowest availability (40.8%) - a disconnect suggesting stock is held but not accessible, possibly due to quality holds
- Chennai and Bangalore have the highest availability rates (54.8% and 56.2%) but are below the 60% target
- Delhi has the fewest SKUs (15) and the lowest total revenue ($81,028) - potentially under-resourced

5.5. Quality & Defects

This sheet contains two linked pivot tables - one grouped by product type, one by supplier - sharing the same formula architecture. It is the most formula-dense sheet in the workbook because splitting inspection outcomes (Pass / Fail / Pending) into separate counts requires COUNTIFS rather than a simple COUNTIF.

Why COUNTIFS Is Required

A native pivot table can count all inspection results in a column, but it cannot split that count into Pass, Fail, and Pending without building three separate calculated fields. COUNTIFS solves this with two criteria: the grouping dimension (product type or supplier name) AND the specific inspection outcome.

Formula Reference - By Product Type

Cell	Formula
`B6`	`=AVERAGEIF(RawData[Product type], A6, RawData[Defect rates]) / 100`
`C6 (Pass)`	`=COUNTIFS(RawData[Product type], A6, RawData[Inspection results], "Pass")`
`D6 (Fail)`	`=COUNTIFS(RawData[Product type], A6, RawData[Inspection results], "Fail")`
`E6 (Pending)`	`=COUNTIFS(RawData[Product type], A6, RawData[Inspection results], "Pending")`
`F6 (Total)`	`=C6 + D6 + E6`
`G6 (Pass Rate)`	`=IFERROR(C6 / F6, 0)`
`H6 (Fail Rate)`	`=IFERROR(D6 / F6, 0)`
`I6 (Pending Rate)`	`=IFERROR(E6 / F6, 0)`

Formula Reference - By Supplier (same pattern, different dimension)

Cell	Formula
`B12`	`=AVERAGEIF(RawData[Supplier name], A12, RawData[Defect rates]) / 100`
`C12 (Pass)`	`=COUNTIFS(RawData[Supplier name], A12, RawData[Inspection results], "Pass")`
`D12 (Fail)`	`=COUNTIFS(RawData[Supplier name], A12, RawData[Inspection results], "Fail")`
`G12 (Pass Rate)`	`=IFERROR(C12 / F12, 0)`

Key Findings

Supplier	SKUs	Pass	Fail	Pending	Pass Rate	Defect Rate
Supplier 1	27	13	6	8	48.1%	1.80%
Supplier 2	22	5	8	9	22.7%	2.36%
Supplier 3	15	2	3	10	13.3%	2.47%
Supplier 4	18	0	12	6	0.0%	2.34%
Supplier 5	18	3	7	8	16.7%	2.67%

- Supplier 4 has zero passing inspections - this is the single most critical quality risk in the dataset
- Supplier 1 is the only supplier with a pass rate above 40% (48.1%), making it the benchmark for quality expectations
- Overall pass rate of 23% across all 100 SKUs indicates a systemic quality issue, not isolated to one supplier

6. Charts & Visual Design

Six embedded charts were built directly from the formula-driven analysis tables. All charts reference analysis table ranges rather than raw data, ensuring they reflect the aggregated (group-level) view rather than individual SKU rows.

Chart	Sheet	Type	Source Range	Purpose
Revenue by Product Type	Revenue by Product	Clustered Column	D3:D6 (revenue)	Revenue comparison across 3 categories
Revenue Share by Product	Revenue by Product	Pie	D3:D6 (revenue)	Proportional revenue contribution
Avg Defect Rate by Supplier	Supplier Performance	Horizontal Bar	F4:F9 (defect %)	Supplier quality ranking
Avg Cost by Transport Mode	Logistics & Shipping	Clustered Column	I5:I9 (avg cost)	Mode cost comparison
Total Revenue by Location	Location & Inventory	Horizontal Bar	F4:F8 (revenue)	Geographic revenue distribution
Defect Rate by Supplier	Quality & Defects	Horizontal Bar	B12:B16 (defect %)	Supplier quality cross-reference

Horizontal bar charts were chosen for supplier and location comparisons because the category labels (supplier names, city names) are long and render more legibly on the vertical axis. Column charts were used for product and mode comparisons where the category labels are short.

7. Conditional Formatting

Conditional formatting was applied strategically across analysis sheets to surface performance patterns without requiring chart interpretation. Three formatting types were used:

Format Type	Applied To	Sheet	Business Signal
3-colour scale (Green→Yellow→Red)	Defect rate column	Supplier Performance	Lowest defect = green; highest = red
3-colour scale (Red→Yellow→Green)	Pass rate column	Supplier Performance	Low pass rate = red (inverse of defect scale)
3-colour scale (Green→Yellow→Red)	Defect rate columns	Quality & Defects	Applied independently to both product and supplier tables
3-colour scale (Red→Yellow→Green)	Pass rate column	Quality & Defects	Inverted so green = high pass rate
3-colour scale (Red→Yellow→Green)	Availability column	Location & Inventory	Red = low availability (stockout risk)
Data bars (blue)	Production volumes	Supplier Performance	Visual proportion of output capacity
Data bars (teal)	Revenue column	Location & Inventory	Relative revenue contribution by city

Colour scales are set to min/max relative to the visible range rather than fixed thresholds, so they adjust automatically as data changes - making the workbook reusable for future datasets without reconfiguration.

8. Reporting Templates

8.1. KPI Report Template

The KPI Report Template is a live dashboard that aggregates outputs from all five analysis sheets into a single summary view. It contains:

- Four headline KPI cards (Total Revenue, Units Sold, Defect Rate, Pass Rate) that pull values via cross-sheet formulas
- A 14-row KPI table covering Financial, Sales, Inventory, Procurement, Logistics, Quality, and Operations categories
- Each KPI row shows: formula/source, current value (live), target, and traffic-light status
- A findings and recommendations section with five pre-written action items based on analysis outcomes

The cross-sheet formula structure used in the KPI cards:

Cell	Formula
`Total Revenue (card)`	`='Revenue by Product'!D8  ← references the totals row`

Units Sold (card)	`='Revenue by Product'!C8`
Defect Rate (card)	`='Quality & Defects'!B9 ← overall average from quality sheet totals row`
Pass Rate (card)	`='Quality & Defects'!G9`
Avg Shipping Time (KPI row)	`='Logistics & Shipping'!C9 ← carrier totals row average`
Revenue per SKU (KPI row)	`='Revenue by Product'!D8 / 'Revenue by Product'!B8`

KPI	Category	Current Value	Target	Status
Total Revenue	Financial	$577,605	$900,000	On Track
Avg Stock Level	Inventory	47.8 units	50+	Below Target
Avg Availability	Inventory	48.4%	60%	Below Target
Avg Lead Time	Procurement	16.0 days	15 days	Above Target
Avg Defect Rate	Quality	2.28%	<2%	Above Target
Inspection Pass Rate	Quality	23.0%	40%	Below Target
Avg Mfg Cost	Operations	$47.27	$40.00	Above Target

8.2. Monthly Report Template

The Monthly Report Template is a structured blank document designed for recurring reporting cycles. It contains eight pre-formatted sections:

1. Executive Summary - free text narrative block
2. Revenue & Sales - metric table with Prior Period / This Period / Target / Variance / Status columns
3. Inventory & Stock - same metric table structure
4. Quality & Defects - same metric table structure
5. Logistics & Shipping - same metric table structure
6. Supplier Performance - same metric table structure
7. Issues & Risks - free text narrative block
8. Actions & Next Steps - free text narrative block

The variance column in each metric table contains a pre-filled formula: =IFERROR(C-B, "") which automatically calculates the difference between This Period and Prior Period values as each row is filled in, reducing manual calculation errors during reporting.

9. Complete Formula Catalogue

The table below documents every distinct formula type used in the workbook, its syntax, purpose, and which sheets it appears in.

Function	Syntax Pattern	Purpose	Sheets Used
SUMIF	=SUMIF(range, criteria, sum_range)	Sum a numeric column for rows matching one criterion	All 5 analysis sheets
AVERAGEIF	=AVERAGEIF(range, criteria, avg_range)	Average a numeric column for rows matching one criterion	All 5 analysis sheets
COUNTIF	=COUNTIF(range, criteria)	Count rows matching one criterion (SKU count per group)	All 5 analysis sheets

COUNTIFS	=COUNTIFS(range1, crit1, range2, crit2)	Count rows matching two criteria simultaneously (supplier + outcome)	Supplier Perf., Quality & Defects
SUM	=SUM(range)	Total row aggregation across all groups	All sheets (totals rows)
AVERAGE	=AVERAGE(range)	Average aggregation for rate metrics in totals rows	All sheets (totals rows)
IFERROR	=IFERROR(formula, fallback)	Suppress #DIV/0! errors in derived ratio columns	Location & Inventory, Quality & Defects
Cross-sheet ref	='Sheet Name'!CellRef	Pull live values from analysis sheets into KPI dashboard	KPI Report Template
Locked ref	=D5/SUM($D$5:$D$7)	Absolute reference locks denominator when formula is copied	Revenue by Product
Division (derived)	=SUMIF(...)/D14	Calculate cost-to-revenue ratio per route	Logistics & Shipping

Total formula count: 337 across all sheets. All formulas reference the named Excel Table (RawData) using structured column references, ensuring compatibility with table expansion.

10. Key Findings & Recommendations

10.1. Critical Issues (Immediate Action)

Issue	Evidence	Recommendation
Supplier 4: zero passing inspections	0 Pass out of 18 SKUs (0% pass rate)	Place Supplier 4 on quality hold; conduct urgent audit before next production run
Systemic inspection failure	Overall pass rate 23% vs 40% target; 36% actively failing	Review quality acceptance criteria across all suppliers; consider third-party inspection
Supplier 5 defect rate	Highest defect rate: 2.67% across 18 SKUs	Formal corrective action plan with measurable improvement milestones

10.2. Performance Gaps (Priority Monitoring)

Gap	Evidence	Recommendation
Low stock availability	Avg 48.4% vs 60% target; Kolkata lowest at 40.8%	Review safety stock policies; investigate Kolkata availability-to-stock disconnect
Lead time overrun	Avg 16.0 days vs 15-day target; Supplier 3 at 20.1 days	Renegotiate Supplier 3 terms or qualify backup supplier
High manufacturing cost	Supplier 4 avg $62.71/unit vs $47.27 overall avg	Given zero pass rate, this cost is not justified; review contract

10.3. Positive Signals

- Skincare maintains strong revenue ($241,628, 41.8%) and a manageable defect rate relative to its SKU volume
- Supplier 1 benchmarks best across quality, cost, and lead time - use as the performance baseline for all supplier negotiations
- Carrier B delivers the best shipping efficiency: most shipments, shortest transit time, second-lowest cost
- Route A provides the best cost-to-volume profile and is the natural default route for high-priority shipments

11. Limitations

- The dataset is synthetic: all figures, supplier names, and locations are simulated. Findings should not be treated as real business intelligence.
- With 100 rows, some groupings are small (Supplier 3: 15 SKUs; Sea mode: 17 shipments). Statistical conclusions from small groups carry higher variance.
- No time dimension is present in the dataset, so trend analysis over periods cannot be performed without additional data.
- The workbook does not use Power Query or pivot tables - it uses SUMIF/AVERAGEIF/COUNTIFS formulas to replicate pivot behaviour. This approach requires manual expansion if new grouping dimensions are added.

12. Conclusion

This project demonstrates a complete Excel analytics workflow from raw data structuring through formula-driven analysis to executive reporting. The 337-formula workbook covers all core Excel data analysis competencies: multi-criteria lookups, derived KPI construction, cross-sheet referencing, conditional formatting, chart embedding, and templated reporting.

The most significant analytical finding is the quality crisis: a 23% pass rate across 100 SKUs, anchored by Supplier 4's zero-pass performance, represents the highest-priority risk in the supply chain. Addressing this single issue through targeted supplier audits and quality controls would have the largest measurable impact on operational reliability.

The workbook architecture - a single named source table driving all downstream formulas and charts - ensures the entire analytical framework updates automatically as new data is loaded, making it a production-ready template for ongoing supply chain monitoring.